Solution to Exercise 2.4.2

We write a straightforward GAP-function which computes the class multiplication matrices M_i for a (small) group G:

```
gap> clmats := function( G )
>    local cl, reps, cle, Mats, i, j , k ;
>    cl := ConjugacyClasses(G);;
>    cle := List( cl, Elements );; reps := List( cl, Representative );;
>    Mats := [];; cle := List( cl, Elements );;
>    for i in [1..Length(cl)] do  Mats[i] := [];
>     for j in [1..Length(cl)] do  Mats[i][j] := [];
>      for k in [1..Length(cl)] do
>       Mats[i][j][k] := Length(
>       Filtered( Cartesian( cle[i],cle[j] ), x -> x[1]*x[2]=reps[k] ) );
>      od;
>     od;
>    od;
>    return( Mats );
>    end;;
```

We next apply this to $G := \mathrm{SL}_2(3)$ and compute the eigenspaces of the class multiplication matrices over the cyclotomic field $K := \mathbb{Q}(\zeta_{12})$. Note that $\exp(G) = 12$.

```
gap> G := SL(2,3);;  Mats = clmats( G );;  K := CyclotomicField(12);;
gap> Eigspaces := List( Mats, A -> Eigenspaces( K, A) );;
gap> DimEigsp := List( Eigspaces, x -> List(x, Dimension) );
[ [ 7 ], [ 1, 1, 1, 1, 1, 1, 1 ], [ 1, 1, 1, 1, 1, 1, 1 ], [ 4, 3 ],
  [ 1, 1, 1, 1, 1, 1, 1 ], [ 1, 1, 1, 1, 1, 1, 1 ], [ 3, 3, 1 ] ]
```

We see that four of the class multiplication matrices have seven pairwise different eigenvalues. So the eigenrow-vectors of any of these matrices are up to scalar multiples the irreducible characters of G. Hence it would suffice to compute just M_i for some $i \in \{2, 3, 5, 6\}$. We choose $i := 2$:

```
gap> A := Mats[2];; Eigvecs := Eigenvectors( K, A );;
gap> cle := List( cl, Elements );;
gap> norms := List( Eigvecs, x -> Sum( List([1..7], i -> Length(cle[i])
>                     * x[i] * ComplexConjugate(x[i]) / Size(G) ) ) );;
gap> Irreds := List( [1..7], i -> ( 1/Sqrt(norms[i]) ) * Eigvecs[i] );
[ [ 1, 1, 1, 1, 1, 1, 1 ], [ 2, 1, 1, -2, -1, -1, 0 ],
  [ 3, 0, 0, 3, 0, 0, -1 ], [ 1, E(3)^2, E(3), 1, E(3), E(3)^2, 1 ],
  [ 2, E(3)^2, E(3), -2, -E(3), -E(3)^2, 0 ],
  [ 1, E(3), E(3)^2, 1, E(3)^2, E(3), 1 ],
  [ 2, E(3), E(3)^2, -2, -E(3)^2, -E(3), 0 ] ]
```

Next we choose $G := \mathrm{V}_4$ and see that here none of the multiplication matrices has four pairwise different eigenvalues:

```
gap> G := DirectProduct( CyclicGroup(2), CyclicGroup(2) );;
gap> Mats := clmats( G );;
gap> Eigspaces := List( Mats, A -> Eigenspaces( K, A) );;
gap> DimEigsp := List( Eigspaces, x -> List(x, Dimension) );
[ [ 4 ], [ 2, 2 ], [ 2, 2 ], [ 2, 2 ] ]
```

Now, let G be an arbitrary finite group. By (2.7) on page 114 we have

$$\mathrm{Irr}_K(G) \cdot M_i^K \cdot \mathrm{Irr}_K(G)^{-1} = \mathrm{diag}(\omega_1(C_i^+), \ldots, \omega_r(C_i^+))$$

where $\mathrm{Irr}_K(G) = [\chi_i(g_j)]_{i,j=1}^r \in K^{r\times r}$, $C_i = g_i^G$ for $i = 1, \ldots, r$ and $\omega_1, \ldots, \omega_r$ are the central characters corresponding to $\chi_1, \ldots, \chi_r$. Since $\omega_i \neq \omega_j$ for $i \neq j$ there is $a = \sum_{i=1}^r \alpha_j C_i^+ \in \mathbf{Z}(KG)$ with $\omega_i(a) \neq \omega_j(a)$ for all $i \neq j$. Hence $\sum_{i=1}^r \alpha_i M_i^K$ has r pairwise different eigenvalues (with multiplicities 1).